

# Towards a miniKanren with fair search strategies

KUANG-CHEN LU, Indiana University  
WEIXI MA, Indiana University  
DANIEL P. FRIEDMAN, Indiana University

We describe fairness levels in disjunction and conjunction implementations. Specifically, a disjunction implementation can be fair, almost-fair, or unfair. And a conjunction implementation can be fair or unfair. We compare the fairness level of four search strategies: the standard miniKanren interleaving depth-first search, the balanced interleaving depth-first search, the fair depth-first search, and the standard breadth-first search. The two non-standard depth-first searches are new. And we present a new, more efficient and simpler implementation of the standard breadth-first search. Using quantitative evaluation, we argue that the two new depth-first searches are competitive alternatives to the standard one, and that our breadth-first search implementation is more efficient than the current one.

## ACM Reference Format:

Kuang-Chen Lu, Weixi Ma, and Daniel P. Friedman. 2019. Towards a miniKanren with fair search strategies. 1, 1 (May 2019), 15 pages. <https://doi.org/10.1145/nnnnnnn.nnnnnnn>

## 1 INTRODUCTION

miniKanren is a family of relational programming languages. [2] Friedman et al. [3] introduce miniKanren and its implementation in *The Reasoned Schemer* and *The Reasoned Schemer, 2nd Ed* (TRS2). Byrd et al. [1] have demonstrated that miniKanren programs are useful in solving several problems. miniKanren.org contains the seeds of many more problems and their solutions.

A subtlety arises when a `conde` contains many clauses: not every clause has an equal chance of contributing to the result. As an example, consider the following relation `repeato` and its invocation.

<sup>c1</sup> LKC: submission deadline: Mon 27 May 2019

<sup>c2</sup> DPF: ... Perhaps it should be part of the title that we have improved on the the standard BFS implementation.

<sup>c3</sup> DPF: When you rewrite the abstract and drop the abbreviations, remember when you are rewriting the Introduction and introducing the abbreviations, that you should explain the ser of BFSser using a reference.

<sup>c4</sup> LKC: I have rewritten the abstract and explain the 'ser' in the ending paragraph of introduction, where we list contributions.

Authors' addresses: Kuang-Chen LuIndiana University; Weixi MaIndiana University; Daniel P. FriedmanIndiana University.

Unpublished working draft. Not for distribution.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from [permissions@acm.org](mailto:permissions@acm.org).

© 2019 Association for Computing Machinery.

XXXX-XXXX/2019/5-ART \$15.00

<https://doi.org/10.1145/nnnnnnn.nnnnnnn>

```

48      (defrel (repeato x out)
49      (conde
50      ((≡ `(,x) out))
51      ((fresh (res)
52      (≡ `(,x . ,res) out)
53      (repeato x res))))))
54
55 > (run 4 q
56     (repeato '* q))
57 '((*) (* *) (* * *) (* * * *))

```

Next, consider the following disjunction of invoking `repeato` with four different letters.

```

60 > (run 12 q
61     (conde
62     ((repeato 'a q))
63     ((repeato 'b q))
64     ((repeato 'c q))
65     ((repeato 'd q))))

```

`conde` intuitively relates its clauses with logical or. And thus an unsuspecting beginner would expect each letter to contribute equally to the result, as follows.

```

69 '((a) (b) (c) (d)
70   (a a) (b b) (c c) (d d)
71   (a a a) (b b b) (c c c) (d d d))

```

The `conde` in TRS2, however, generates a less expected result.

```

74 '((a) (a a) (b) (a a a)
75   (a a a a) (b b)
76   (a a a a a) (c)
77   (a a a a a a) (b b b)
78   (a a a a a a a) (d))
79

```

The miniKanren in TRS2 implements interleaving DFS (DFS<sub>i</sub>), the cause of this unexpected result. With this search strategy, each `conde` clause takes half of its received computational resources and passes the other half to its following clauses, except for the last clause that takes all resources it receives. In the example above, the a clause takes half of all resources. And the b clause takes a quarter. Thus c and d barely contribute to the result.

DFS<sub>i</sub> is sometimes powerful for an expert. By carefully organizing the order of `conde` clauses, a miniKanren program can explore more “interesting” clauses than those uninteresting ones, and thus use computational resources efficiently.

DFS<sub>i</sub> is not always the best choice. For instance, it might be less desirable for little miniKanreners – understanding implementation details and fiddling with clause order is not their first priority. There is another reason that miniKanren could use more search strategies than just DFS<sub>i</sub>. In many applications, there does not exist one order that serves all purposes. For example, a relational dependent type checker contains clauses for constructors that build data and clauses for eliminators that use data. When the type checker is generating simple and shallow programs, the clauses for constructors had better be at the top of the `conde` expression. When performing proof searches for complicated programs, the clauses for eliminators had better be at the top of the `conde` expression.

With  $\text{DFS}_i$ , these two uses cannot be efficient at the same time. In fact, to make one use efficient, the other one must be more sluggish.

The specification that gives every clause in the same  $\text{cond}^e$  equal “search priority” is fair  $\text{disj}$ . And search strategies with almost-fair  $\text{disj}$  give every clause similar priority. Fair  $\text{conj}$ , a related concept, is more subtle. We cover it in the next section.

To summarize our contributions, we

- propose and implement **balanced interleaving** depth-first search ( $\text{DFS}_{bi}$ ), a new search strategy with almost-fair  $\text{disj}$ .
- propose and implement **fair** depth-first search ( $\text{DFS}_f$ ), a new search strategy with fair  $\text{disj}$ .
- implement in a new way the standard breath-first search (BFS), a search strategy with fair  $\text{disj}$  and fair  $\text{conj}$ . We name the current implementation by Seres et al. as  $\text{BFS}_{ser}$ , and our new implementation as  $\text{BFS}_{imp}$ . We prove formally that they are semantically equivalent.  $\text{BFS}_{imp}$ , however, runs faster in all benchmarks and is shorter, thus being an **improvement**.

## 2 SEARCH STRATEGIES AND FAIRNESS

In this section, we define fair  $\text{disj}$ , almost-fair  $\text{disj}$  and fair  $\text{conj}$ . Before going further into fairness, we give a short review of the terms: *state*, *search space*, and *goal*. A *state* is a collection of constraints. (Here, we restrict constraints to unification constraints.) Every answer corresponds to a state. A *space* is a collection of states. And a *goal* is a function from a state to a space.

Now we elaborate fairness by running more queries about  $\text{repeat}^0$ . We never use  $\text{run}^*$  in this paper because fairness is more interesting when we have an unbounded number of answers. However, it is perfectly fine to use  $\text{run}^*$  with any search strategies.

### 2.1 fair $\text{disj}$

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the query result.  $\text{DFS}_i$ , the current search strategy, however, results in many more lists of *a* than lists of other letters. And some letters (e.g. *c* and *d*) are rarely seen. The situation would be exacerbated if the  $\text{cond}^e$  would have contained more clauses.

```
;;  $\text{DFS}_i$  (unfair  $\text{disj}$ )
> (run 12 q
  (conde
    ((repeat0 'a q))
    ((repeat0 'b q))
    ((repeat0 'c q))
    ((repeat0 'd q))))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

<sup>c1</sup>*LKC*: The following paragraph is extended to include an explanation on  $\text{run}^*$

Under the hood, the `conde` here is allocating computational resources to four trivially different search spaces. The unfair `disj` in `DFSi` allocates many more resources to the first search space. On the contrary, fair `disj` would allocate resources evenly to each search space.

<pre> 142 ;; DFS<sub>f</sub> (fair disj) 143 &gt; (run 12 q 144   (cond<sup>e</sup> 145     ((repeat<sup>o</sup> 'a q)) 146     ((repeat<sup>o</sup> 'b q)) 147     ((repeat<sup>o</sup> 'c q)) 148     ((repeat<sup>o</sup> 'd q)))) 149 '((a) (b) (c) (d) 150   (a a) (b b) (c c) (d d) 151   (a a a) (b b b) (c c c) (d d d)) </pre>	<pre> 142 ;; BFS (fair disj) 143 &gt; (run 12 q 144   (cond<sup>e</sup> 145     ((repeat<sup>o</sup> 'a q)) 146     ((repeat<sup>o</sup> 'b q)) 147     ((repeat<sup>o</sup> 'c q)) 148     ((repeat<sup>o</sup> 'd q)))) 149 '((a) (b) (c) (d) 150   (a a) (b b) (c c) (d d) 151   (a a a) (b b b) (c c c) (d d d)) </pre>
---	---

Running the same program again with almost-fair `disj` (e.g. `DFSbi`) gives the same result. Almost-fair, however, is not completely fair, as shown by the following example.

```

159 ;; DFSbi (almost-fair disj)
160 > (run 16 q
161   (conde
162     ((repeato 'a q))
163     ((repeato 'b q))
164     ((repeato 'c q))
165     ((repeato 'd q))
166     ((repeato 'e q))))
167 '((b) (c) (d) (a)
168   (b b) (c c) (d d) (e)
169   (b b b) (c c c) (d d d) (a a)
170   (b b b b) (c c c c) (d d d d) (e e))

```

`DFSbi` is fair only when the number of goals is a power of 2, otherwise, it allocates some goals twice as many resources as the others. In the above example, where the `conde` has five clauses, `DFSbi` allocates more resources to the clauses of b, c, and d.

We end this subsection with precise definitions of all levels of `disj` fairness. Our definition of *fair disj* is slightly more general than the one in Seres et al. [6]. Their definition is only for binary disjunction. We generalize it to a multi-arity one.

**DEFINITION 2.1 (FAIR `disj`).** *A `disj` is fair if and only if it allocates computational resources evenly to search spaces produced by goals in the same disjunction (i.e., clauses in the same `conde`).*

**DEFINITION 2.2 (ALMOST-FAIR `disj`).** *A `disj` is almost-fair if and only if it allocates computational resources so evenly to search spaces produced by goals in the same disjunction that the maximal ratio of resources is bounded by a constant.*

**DEFINITION 2.3 (UNFAIR `disj`).** *A `disj` is unfair if and only if it is not almost-fair.*

## 2.2 fair conj

c1

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the answer list. Search strategies with unfair conj (e.g.  $\text{DFS}_i$ ,  $\text{DFS}_{bi}$ ,  $\text{DFS}_f$ ), however, results in many more lists of a than lists of other letters. And some letters are rarely seen. The situation would be exacerbated if  $\text{cond}^e$  were to contain more clauses. Although some strategies have a different level of fairness in  $\text{disj}$ , they have the same behavior when there is no call to a relational definition in  $\text{cond}^e$  clauses (including this case).

<pre>;; DFS<sub>i</sub> (unfair conj) &gt; (run 12 q    (fresh (x)     (cond<sup>e</sup>       ((≡ 'a x))       ((≡ 'b x))       ((≡ 'c x))       ((≡ 'd x)))     (repeat<sup>o</sup> x q))) '((a) (a a) (b) (a a a)   (a a a a) (b b)   (a a a a a) (c)   (a a a a a a) (b b b)   (a a a a a a a) (d))</pre>	<pre>;; DFS<sub>f</sub> (unfair conj) &gt; (run 12 q    (fresh (x)     (cond<sup>e</sup>       ((≡ 'a x))       ((≡ 'b x))       ((≡ 'c x))       ((≡ 'd x)))     (repeat<sup>o</sup> x q))) '((a) (a a) (b) (a a a)   (a a a a) (b b)   (a a a a a) (c)   (a a a a a a) (b b b)   (a a a a a a a) (d))</pre>	<pre>;; DFS<sub>bi</sub> (unfair conj) &gt; (run 12 q    (fresh (x)     (cond<sup>e</sup>       ((≡ 'a x))       ((≡ 'b x))       ((≡ 'c x))       ((≡ 'd x)))     (repeat<sup>o</sup> x q))) '((a) (a a) (c) (a a a)   (a a a a) (c c)   (a a a a a) (b)   (a a a a a a) (c c c)   (a a a a a a a) (d))</pre>
---	---	--

Under the hood, the  $\text{cond}^e$  and the call to  $\text{repeat}^o$  are connected by conj. The  $\text{cond}^e$  goal outputs a search space including four trivially different states. Applying the next conjunctive goal,  $(\text{repeat}^o \ x \ q)$ , produces four trivially different search spaces. In the examples above, all search strategies allocate more computational resources to the search space of a. On the contrary, fair conj would allocate resources evenly to each search space. For example,

```
;; BFS (fair conj)
> (run 12 q
   (fresh (x)
    (conde
      ((≡ 'a x))
      ((≡ 'b x))
      ((≡ 'c x))
      ((≡ 'd x)))
    (repeato x q)))
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```

A more interesting situation is when the first conjunct produces an unbounded number of states. Consider the following example, a naive specification of fair conj might require search strategies to produce all sorts of singleton lists, but there would not be any lists of length two or longer, which makes the strategies incomplete. A

<sup>c1</sup> LKC: I change '1/4 in the answer' to '1/4 in the answer list'.

search strategy is *complete* if and only if “every correct answer would be discovered after some finite time” [6], otherwise, it is *incomplete*. In the context of miniKanren, a search strategy is complete means that every correct answer has a position in large enough answer lists.

<sup>c1</sup> <sup>c2</sup> <sup>c3</sup>

```
;; naively fair conj
> (run 6 q
  (fresh (xs)
    (conde
      ((repeato 'a xs))
      ((repeato 'b xs))
      (repeato xs q)))
  '(((a)) ((b))
    ((a a)) ((b b))
    ((a a a)) ((b b b))))
```

Our solution requires a search strategy with *fair* conj to organize states in bags in search spaces, where each bag contains finite states, and to allocate resources evenly among search spaces derived from states in the same bag. It is up to a search strategy designer to decide by what criteria to put states in the same bag, and how to allocate resources among search spaces related to different bags.

BFS puts states of the same cost in the same bag, and allocates resources carefully among search spaces related to different bags such that it produces answers in increasing order of cost. The *cost* of an answer is its depth in the search tree (i.e., the number of calls to relations required to find them) [6]. In the following example, every answer is a list of a list of symbols, where inner lists in the same outer list are identical. Here the cost of each answer is equal to the length of its inner list plus the length of its outer list. For example, the cost of ((a) (a)) is  $1 + 2 = 3$ .

<sup>c4</sup> <sup>c5</sup>

<sup>c1</sup>DPF: I think that should be said as directly as in your note.

<sup>c2</sup>LKC: I state this fact at the beginning of this section.

<sup>c3</sup>LKC: We never use run\* in this paper because fairness is only interesting when we have an unbounded number of answers. However, it is perfectly fine to use run\* with any search strategies.

<sup>c4</sup>DPF: Next to the last line of the above paragraph the word same has too many meanings. It could be have the same cost or it could be failing to distinguish two list that would be the same if we substituted a b for an a. It needs to be clarified. It might help to work out the actual numbers

<sup>c5</sup>LKC: I have rewritten the relevant sentences.

```

283 ;; BFS (fair conj)
284 > (run 12 q
285   (fresh (xs)
286    (conde
287     ((repeato 'a xs))
288     ((repeato 'b xs)))
289    (repeato xs q)))
290 '(((a)) ((b)))
291 ((a) (a)) ((b) (b))
292 ((a a)) ((b b))
293 ((a) (a) (a)) ((b) (b) (b))
294 ((a a) (a a)) ((b b) (b b))
295 ((a a a)) ((b b b))
296

```

We end this subsection with precise definitions of all levels of conj fairness.

DEFINITION 2.4 (FAIR conj). *A conj is fair if and only if it allocates computational resources evenly to search spaces produced from states in the same bag. A bag is a finite collection of states. And search strategies with fair conj should represent search spaces with possibly unbounded collections of bags.*

DEFINITION 2.5 (UNFAIR conj). *A conj is unfair if and only if it is not fair.*

### 3 INTERLEAVING DEPTH-FIRST SEARCH

In this section, we review the implementation of interleaving depth-first search (DFS<sub>i</sub>). We focus on parts that are relevant to this paper. TRS2, chapter 10 and the appendix, “Connecting the wires”, provides a comprehensive description of the miniKanren implementation but limited to unification constraints ( $\equiv$ ). Fig. 1 and Fig. 2 show parts that are later compared with other search strategies. We follow some conventions to name variables: ss name states; gs (possibly with subscript) name goals; variables ending with  $^\infty$  name search spaces. Fig. 1 shows the implementation of disj. The first function, disj2, implements binary disjunction. It applies the two disjunctive goals to the input state s and composes the two resulting search spaces with append $^\infty$ . The following syntax definitions say disj is right-associative. Fig. 2 shows the implementation of conj. The first function, conj2, implements binary conjunction. It applies the *first* goal to the input state, then applies the second goal to states in the resulting search space. The helper function append-map $^\infty$  applies its input goal to states in its input search spaces and composes the resulting search spaces. It reuses append $^\infty$  for search space composition. The following syntax definitions say conj is also right-associative.

<sup>c1</sup> c2

### 4 BALANCED INTERLEAVING DEPTH-FIRST SEARCH

Balanced interleaving DFS (DFS<sub>bi</sub>) has an almost-fair disj and unfair conj. The implementation of DFS<sub>bi</sub> differs from DFS<sub>i</sub>’s in the disj macro. We list the new disj with its helper in Fig. 3. When there are one or more disjunctive goals, disj builds a balanced binary tree whose leaves are the goals and whose nodes are disj2s, hence the name of this search strategy. The new helper, disj+, takes two additional ‘arguments’. They accumulate goals to be put in the left and right subtrees. The first clause handles the case where there is only one goal. In this case, the tree is the goal itself. When there are more goals, we partition the list of goals into two sublists of

<sup>c1</sup>DPF: This came out fabulous. I think I spotted some missed opportunities later, which I will point out with additional notes.

<sup>c2</sup>LKC: I think we can simply remove cond<sup>e</sup> code and its accompany description.

```

330 #| Goal × Goal → Goal |#
331 (define (disj2 g1 g2)
332   (lambda (s)
333     (append∞ (g1 s) (g2 s))))
334
335 #| Space × Space → Space |#
336 (define (append∞ s∞ t∞)
337   (cond
338     ((null? s∞) t∞)
339     ((pair? s∞)
340      (cons (car s∞)
341            (append∞ (cdr s∞) t∞)))
342     (else (lambda ()
343              (append∞ t∞ (s∞))))))
344
345
346 (define-syntax disj
347   (syntax-rules ()
348     ((disj) (fail))
349     ((disj g0 g ...) (disj+ g0 g ...))))
350
351 (define-syntax disj+
352   (syntax-rules ()
353     ((disj+ g) g)
354     ((disj+ g0 g1 g ...) (disj2 g0 (disj+ g1 g ...))))
355
356

```

Fig. 1. implementation of DFS<sub>i</sub> (Part I)

roughly equal lengths and recur on the two sublists. We move goals to the accumulators in the last clause. As we are moving two goals each time, there are two base cases: (1) no goal remains; (2) one goal remains. We handle these two new base cases in the second clause and the third clause, respectively. In contrast, the `disj` in DFS<sub>i</sub> constructs the binary tree in a particularly unbalanced form.

## 5 FAIR DEPTH-FIRST SEARCH

Fair DFS (DFS<sub>f</sub>) has fair `disj` and unfair `conj`. The implementation of DFS<sub>f</sub> differs from DFS<sub>i</sub>'s in `disj2` (Fig. 4). The new `disj2` calls a new and fair version of `append∞`. `append∞fair` immediately calls its helper, `loop`, with the first argument, `s?`, set to `#t`, which indicates that we haven't swapped `s∞` and `t∞`. The swapping happens at the third `cond` clause in the helper, where `s?` is updated accordingly. The first two `cond` clauses essentially copy the cars and stop recursion when one of the input spaces is obviously finite. The third clause, as we mentioned above, is just for swapping. When the fourth and last clause runs, we know that both `s∞` and `t∞` are ended with a thunk, and that we have swapped them. In this case, we construct a new thunk. The new thunk swaps two



```

377 #| Goal × Goal → Goal |#
378 (define (conj2 g1 g2)
379   (lambda (s)
380     (append-map∞ g2 (g1 s))))
381
382 #| Goal × Space → Space |#
383 (define (append-map∞ g s∞)
384   (cond
385     ((null? s∞) '())
386     ((pair? s∞)
387      (append∞ (g (car s∞))
388                (append-map∞ g (cdr s∞))))
389     (else (lambda ()
390              (append-map∞ g (s∞))))))
391
392
393 (define-syntax conj
394   (syntax-rules ()
395     ((conj) (fail))
396     ((conj g0 g ...) (conj+ g0 g ...))))
397
398 (define-syntax conj+
399   (syntax-rules ()
400     ((conj+ g) g)
401     ((conj+ g0 g1 g ...) (conj2 g0 (conj+ g1 g ...))))
402
403

```

Fig. 2. implementation of DFS<sub>i</sub> (Part II)

spaces back in the recursive call to loop. This is unnecessary for fairness. We do it to produce answers in a more natural order.

<sup>c1</sup> <sup>c2</sup>

## 6 BREADTH-FIRST SEARCH

BFS has both fair disj and fair conj. An easy implementation is based on DFS<sub>f</sub> (not DFS<sub>i</sub>): (1) replace append<sup>∞</sup> with append<sup>∞</sup><sub>fair</sub> in append-map<sup>∞</sup>'s body (2) rename append-map<sup>∞</sup> to append-map<sup>∞</sup><sub>fair</sub>.

<sup>c1</sup>

This implementation is improvable in two ways. First, as mentioned in subsection 2.2, BFS puts answers in bags and answers of the same cost are in the same bag. In this implementation, however, we manage cost information subtly—the cars of a search space have cost 0 (i.e. they are all in the same bag), and every thunk indicates an increment in cost. It is even more subtle that append<sup>∞</sup><sub>fair</sub> and the append-map<sup>∞</sup><sub>fair</sub> respects the cost information.

<sup>c1</sup> DPF: When or where do you swap them back? This could be still clearer.

<sup>c2</sup> LKC: Improved.

<sup>c1</sup> LKC: Applying the above changes won't result in BFS<sub>ser</sub>, but yet another implementation of BFS.

```

424 (define-syntax disj
425   (syntax-rules ()
426     ((disj) fail)
427     ((disj g ...) (disj+ (g ...) () ())))))
428
429 (define-syntax disj+
430   (syntax-rules ()
431     ((disj+ () () g) g)
432     ((disj+ (gl ...) (gr ...))
433      (disj2 (disj+ () () gl ...)
434              (disj+ () () gr ...)))
435     ((disj+ (gl ...) (gr ...) g0)
436      (disj2 (disj+ () () gl ... g0)
437              (disj+ () () gr ...)))
438     ((disj+ (gl ...) (gr ...) g0 g1 g ...)
439      (disj+ (gl ... g0) (gr ... g1) g ...)))
440
441
442

```

Fig. 3. implementation of DFS<sub>bi</sub>

```

445 #| Goal × Goal → Goal |#
446 (define (disj2 g1 g2)
447   (lambda (s)
448     (append∞fair (g1 s) (g2 s))))
449
450 #| Space × Space → Space |#
451 (define (append∞fair s∞ t∞)
452   (let loop ((s? #t) (s∞ s∞) (t∞ t∞))
453     (cond
454       ((null? s∞) t∞)
455       ((pair? s∞)
456        (cons (car s∞)
457              (loop s? (cdr s∞) t∞)))
458       (s? (loop #f t∞ s∞))
459       (else (lambda ()
460                (loop #t (t∞) (s∞)))))))
461
462
463

```

Fig. 4. implementation of DFS<sub>f</sub>

Second,  $\text{append}_{\text{fair}}^{\infty}$  is extravagant in memory usage. It makes  $O(n + m)$  new cons cells every time, where  $n$  and  $m$  are the “length”s of input search spaces.

<sup>c2</sup> LKC: Should I point out that DFS<sub>f</sub> also has the miserable space extravagance?

In the following subsections, we firstly describe  $\text{BFS}_{\text{imp}}$ , an improved BFS implementation which manages cost information in a more clear way and is less extravagant in memory usage. Then we compare  $\text{BFS}_{\text{imp}}$  with  $\text{BFS}_{\text{ser}}$ , the current implementation by Seres et al..

## 6.1 improved BFS

We make the cost information more clear by changing the type of search space, modifying related function definitions, and introducing a few more functions.

The new type is a pair whose *car* is a list of answers (the bag), and whose *cdr* is either  $\#f$  or a thunk returning a search space. A falsy *cdr* means the search space is obviously finite.

We list functions related to the pure subset in Fig. 5. The first three functions are search space constructors. *none* makes an empty search space; *unit* makes a space from one answer; and *step* makes a space from a thunk. The remaining functions are as before. We compare these functions with  $\text{BFS}_{\text{ser}}$  in our proof.

Luckily, the change in  $\text{append}_{\text{fair}}^\infty$  also fixes the miserable space extravagance—the use of *append* helps us to reuse the first bag of  $t^\infty$ .

Kiselyov et al. [4] have demonstrated that a *MonadPlus* hides in implementations of logic programming system.  $\text{BFS}_{\text{imp}}$  is not an exception:  $\text{append-map}_{\text{fair}}^\infty$  is like *bind*, but takes arguments in reversed order; *none*, *unit*, and  $\text{append}_{\text{fair}}^\infty$  correspond to *mzero*, *unit*, and *mplus*, respectively.

Functions implementing impure features are in Fig. 6. The first function, *elim*, takes a space  $s^\infty$  and two continuations *kf* and *ks*. When  $s^\infty$  contains no answers, it calls *kf* with no argument. Otherwise, it calls *ks* with the first answer and the rest of the space. Here ‘f’ means ‘fail’ and ‘s’ means ‘succeed’. This function is similar to an eliminator of search spaces, hence the name. The remaining functions are as before.

## 6.2 compare $\text{BFS}_{\text{imp}}$ with $\text{BFS}_{\text{ser}}$

In this subsection, we compare the pure subset of  $\text{BFS}_{\text{imp}}$  with  $\text{BFS}_{\text{ser}}$ . We focus on the pure subset because  $\text{BFS}_{\text{ser}}$  is designed for a pure relational programming system. We proof in Coq that these two search strategies are semantically equivalent, (i.e.,  $(\text{run } n \ ? \ g)$  produces the same result (See supplements for the formal proof). To compare efficiency, we translate  $\text{BFS}_{\text{ser}}$ ’s Haskell code into Racket (See supplements for the translated code). The translation is direct due to the similarity in two relational programming systems. The translated code is longer than  $\text{BFS}_{\text{imp}}$ . And it runs slower in all benchmarks. Details about differences in efficiency are in Table 1.

## 7 QUANTITATIVE EVALUATION

In this section, we compare the efficiency of search strategies. A concise description is in Table 1. A hyphen means “running out of memory.” The first two benchmarks are from TRS2. *revers<sup>o</sup>* is from Rozplokh and Boulytchev [5]. The next two benchmarks about quine are modified from a similar test case in Byrd et al. [1]. The modifications are made to circumvent the need for symbolic constraints (e.g.  $\neq$ , *absent<sup>o</sup>*). Our version generates de Bruijnized expressions and prevents closures from being inside a list. The two benchmarks differ in the *cond<sup>e</sup>* clause order of their relational interpreters. The last two benchmarks are about synthesizing expressions that evaluate to ‘(I love you)’. This benchmark is from Byrd et al. [1]. Again, the sibling benchmarks differ in the

<sup>c1</sup> LKC: I do notice that in the below paragraph the leading sentence is not at the beginning (but at the second position). Putting the current first sentence at the beginning, however, looks more fun to me.

<sup>c2</sup> LKC: I swap *ks* and *kf* to mimic *ind-List*

<sup>c3</sup> LKC: I have rewritten this subsection.

```

518 #|  $\rightarrow \text{Space}$  |#
519 (define (none) `(( . #f))
520
521 #|  $\text{State} \rightarrow \text{Space}$  |#
522 (define (unit s) `((,s) . #f))
523
524 #|  $(\rightarrow \text{Space}) \rightarrow \text{Space}$  |#
525 (define (step f) `(( . ,f))
526
527
528 #|  $\text{Space} \times \text{Space} \rightarrow \text{Space}$  |#
529 (define (append∞fair s∞ t∞)
530   (cons (append (car s∞) (car t∞))
531         (let ((t1 (cdr s∞)) (t2 (cdr t∞)))
532           (cond
533             ((not t1) t2)
534             ((not t2) t1)
535             (else (lambda ()
536                     (append∞fair (t1) (t2)))))))
537
538
539 #|  $\text{Goal} \times \text{Space} \rightarrow \text{Space}$  |#
540 (define (append-map∞fair g s∞)
541   (foldr
542     (lambda (s t∞)
543       (append∞fair (g s) t∞))
544     (let ((f (cdr s∞)))
545       (step (and f (lambda () (append-map∞fair g (f)))))
546     (car s∞)))
547
548
549 #|  $\text{Maybe Nat} \times \text{Space} \rightarrow [\text{State}]$  |#
550 (define (take∞ n s∞)
551   (let loop ((n n) (vs (car s∞)))
552     (cond
553       ((and n (zero? n)) '())
554       ((pair? vs)
555        (cons (car vs)
556              (loop (and n (sub1 n)) (cdr vs))))
557       (else
558        (let ((f (cdr s∞)))
559          (if f (take∞ n (f)) '())))))
560
561

```

Fig. 5. new and changed functions in optimized BFS that implements pure features

```

565 #| Space × (State × Space → Space) × (→ Space) → Space |#
566 (define (elim s∞ kf ks)
567   (let ((ss (car s∞)) (f (cdr s∞)))
568     (cond
569       ((and (null? ss) f)
570        (step (lambda () (elim (f) kf ks))))
571       ((null? ss) (kf))
572       (else (ks (car ss) (cons (cdr ss) f))))))
573
574
575 #| Goal × Goal × Goal → Goal |#
576 (define (ifte g1 g2 g3)
577   (lambda (s)
578     (elim (g1 s)
579           (lambda () (g3 s))
580           (lambda (s0 s∞)
581             (append-map∞fair g2
582              (append∞fair (unit s0) s∞))))))
583
584
585 #| Goal → Goal |#
586 (define (once g)
587   (lambda (s)
588     (elim (g s)
589           (lambda () (none))
590           (lambda (s0 s∞) (unit s0))))))
591

```

Fig. 6. new and changed functions in improved BFS that implement impure features

cond<sup>e</sup> clause order of their relational interpreters. The first one has elimination rules (i.e. application, car, and cdr) at the end, while the other has them at the beginning. We conjecture that DFS<sub>i</sub> would perform badly in the second case because elimination rules complicate the problem when synthesizing (i.e., our evaluation supports our conjecture.)

In general, only DFS<sub>i</sub> and DFS<sub>bi</sub> constantly perform well. DFS<sub>f</sub> is just as efficient in all benchmarks but very-recursive<sup>o</sup>. Both implementations of BFS have obvious overhead in many cases. Among the three variants of DFS, which all have unfair conj, DFS<sub>f</sub> is most resistant to clause permutation, followed by DFS<sub>bi</sub> then DFS<sub>i</sub>. Among the two implementation of BFS, our improved BFS<sub>imp</sub> constantly performs as well or better. Interestingly, every strategy with fair disj suffers in very-recursive<sup>o</sup> and DFS<sub>f</sub> performs well elsewhere. Therefore, this benchmark might be a special case. Fair conj imposes considerable overhead constantly except in append<sup>o</sup>. The reason might be that strategies with fair conj tend to keep more intermediate answers in the memory.

## 8 RELATED WORKS

Yang [7] points out that a disjunct complex would be ‘fair’ if it were a full and balanced tree.

benchmark	size	DFS <sub>i</sub>	DFS <sub>bi</sub>	DFS <sub>f</sub>	BFS <sub>imp</sub>	BFS <sub>ser</sub>
very-recursive <sup>o</sup>	100000	579	793	2131	1438	3617
	200000	1283	1610	3602	2803	4212
	300000	2160	2836	-	6137	-
append <sup>o</sup>	100	31	41	42	31	68
	200	224	222	221	226	218
	300	617	634	593	631	622
revers <sup>o</sup>	10	5	3	3	38	85
	20	107	98	51	4862	5844
	30	446	442	485	123288	132159
quine-1	1	71	44	69	-	-
	2	127	142	95	-	-
	3	114	114	93	-	-
quine-2	1	147	112	56	-	-
	2	161	123	101	-	-
	3	289	189	104	-	-
'(I love you)-1	99	56	15	22	74	165
	198	53	72	55	47	74
	297	72	90	44	181	365
'(I love you)-2	99	242	61	16	66	99
	198	445	110	60	42	64
	297	476	146	49	186	322

Table 1. The results of a quantitative evaluation: running times of benchmarks in milliseconds

Seres et al. [6] describe a breadth-first search strategy. We present another implementation. Our implementation is semantically equivalent to theirs. But, ours is shorter and performs better in comparison with a straightforward translation of their Haskell code.

## 9 CONCLUSION

We analyze the definitions of fair disj and fair conj, then propose a new definition of fair conj. Our definition is orthogonal with completeness.

We devise two new search strategies (i.e. balanced interleaving DFS (DFS<sub>bi</sub>) and fair DFS (DFS<sub>f</sub>)) and devise a new implementation of BFS. These strategies have different features in fairness: DFS<sub>bi</sub> has an almost-fair disj and unfair conj. DFS<sub>f</sub> has fair disj and unfair conj. BFS has both fair disj and fair conj.

Our quantitative evaluation shows that DFS<sub>bi</sub> and DFS<sub>f</sub> are competitive alternatives to DFS<sub>i</sub>, the current search strategy, and that BFS is less practical.

Our new implementation of BFS is semantically equivalent to the original one. But, ours is shorter and performs better in comparison with a straightforward translation of their Haskell code.

<sup>c1</sup> <sup>c2</sup>

<sup>c1</sup>DPF: Finish this paragraph with a clarification and you may place this anywhere in the conclusion. This may require some consultation with Weixi.

<sup>c2</sup>LKC: I will do it later.

## ACKNOWLEDGMENTS

## REFERENCES

- [1] William E Byrd, Michael Ballantyne, Gregory Rosenblatt, and Matthew Might. 2017. A unified approach to solving seven programming problems (functional pearl). *Proceedings of the ACM on Programming Languages* 1, ICFP (2017).
- [2] Daniel P. Friedman, William E. Byrd, and Oleg Kiselyov. 2005. *The Reasoned Schemer*. The MIT Press.
- [3] Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition*.
- [4] Oleg Kiselyov, Chung-chieh Shan, Daniel P Friedman, and Amr Sabry. 2005. Backtracking, interleaving, and terminating monad transformers:(functional pearl). *ACM SIGPLAN Notices* 40, 9 (2005), 192–203.
- [5] Dmitri Rozplokhas and Dmitri Boulytchev. 2018. Improving Refutational Completeness of Relational Search via Divergence Test. In *Proceedings of the 20th International Symposium on Principles and Practice of Declarative Programming*. ACM, 18.
- [6] Silvija Seres, J Michael Spivey, and C. A. R. Hoare. 1999. Algebra of Logic Programming.. In *ICLP*. 184–199.
- [7] Edward Z. Yang. 2010. Adventures in Three Monads. *The Monad. Reader Issue* 15 (2010), 11.